UNIVERSITETET I OSLO
UNIVERSITY OF OSLO

CTF.BAS Ver 1.1

A BASIC program for an IBM PC compatible computer for drawing the weak phase object contrast transfer function.

by

Arne Olsen and Per Skjerpe

Department of Physics, University of Oslo
P.O. Box 1048, 0316 OSLO 3

ovr--

Report 89-25 Received: 1989-12-11

ISSN-0332-5571

FYSISK INSTITUTT
DEPARTMENT OF PHYSICS



ABSTRACT

This report describes a program, for an IBM PC compatible computer, which is useful in high resolution electron microscopy (HREM). The program, called CTF.BAS Ver 1.1, is written in EBASIC and calculates the weak phase object contrast transfer function as function of instrumental and imaging parameters. The function is plotted on the PC graphics screen and by a Print Screen command the function can be copied to the printer. The present program version runs on both the Hercules graphic card and the IBM Color Graphics Adapter.

1. INTRODUCTION

In high resolution electron microscopy (HREM), the weak phase object approximation is commonly used for the interpretation of experimental images although its range of validity is restricted. In this approximation, the effect of instrumental and imaging parameters on the high resolution image can be represented by a contrast transfer function:

$$I(g) = 2\sigma V_g \, t \, A(g)$$

Here I(g) is the Fourier coefficient of the HREM image and

σ = electron interaction constant
V_g = Fourier coefficient (Volt)
t = specimen thickness
A(u) = weak phase object contrast transfer function (CTF)

The first term, $\sigma V_g t$, represents the specimen structure and the last term, A(g), represents the distortion in the imaging system. For the experimentalist it is important to gain control over the last term.

The present report presents a BASIC program which plots the weak phase object transfer function A(u) as function of various parameters.

2. HOW TO RUN THE PROGRAM

Insert programme diskette in drive A: and type:

A: <ENTER>

Start program by typing:

EBASIC CTF <ENTER>

When the plot has been completed, the machine makes a beep and waits for any key to be pressed before continuing. If the pattern is to be copied to printer, do it now (before pressing other keys).

3. THEORY

The defocus setting and the spherical aberration of the objective lens impose a phase shift on the reciprocal space electron wave function. This well known instrumental phase shift is:

$$A(u) = \exp(iX(u))$$

where: $X(u) = \pi f \lambda u^2 + \pi C_s \lambda^3 u^4/2$

u = reciprocal space coordinate (inverse length);
f = defocus (negative for underfocus);
λ = electron wavelength;
C_s = spherical aberration constant;

In addition comes the damping effect of chromatical aberration. This is caused by instrumental instabilities in the accelerating voltage and the lens current, and also the energy spread in electrons leaving the filament comes into consideration. Including these damping effects, the transfer function becomes:

$$A(u) = \exp(iX(u)) \exp(-\pi^2 \, del^2 \, \lambda^2 \, u^4/2)$$

where: $del = C_c \sqrt{(dV/V)^2 + 4(dI/I)^2 + (dE/E)^2}$

C_c = chromatical aberration constant;
(dV/V) = accelerating voltage fluctation;
(dI/I) = lens current fluctuation;
(dE/E) = electron energy spread;

Another damping effect arises due to the angular spread in the illuminating electron beam. Including this damping effect the transfer function becomes:

$$A(u) = \exp(iX(u)) \exp(-\pi^2 \, del^2 \, \lambda^2 \, u^4/2) \exp(-\pi^2 \, u_c^2 \, q^2)$$

where: $q = C_s \lambda^3 u^3 + f \lambda u$ and $u_c = \theta_c / \sqrt{\ln 2} \, \lambda$

θ_c = convergence semiangle (radians)

For a weak phase object, it can be shown that the Fourier coefficient $I(g)$ of the high resolution electron image is:

$$I(g) = 2\sigma V_g \, t \, A(u)$$

where
σ = electron interaction constant
V_g = Fourier coefficient (Volt)
t = specimen thickness

and where the weak phase object contrast transfer function (CTF) is determined by the imaginary part of the transfer function:

$$A(u) = \sin(X(u)) \exp(-\pi^2 \, del^2 \lambda^2 u^4/2) \exp(-\pi^2 u_c^2 \, q^2)$$

This is the actual expression used by program CTF.BAS

APPENDIX 1. Source listing of program CTF.BAS Ver 1.1

```
1000 '****************************************
1010 '          PROGRAM CTF.BAS  Ver 1.1
1020 ' Plot contrast transfer function on screen
1030 '         Written in EBASIC ver 3.0
1040 '                  by
1050 '    Arne Olsen and Per Skjerpe  1989
1060 '****************************************
1070 '
1080 CLS:PRINT " Program CTF  Ver 1.1 ":PRINT
1090 PRINT "Indicate graphics card:"
1100 PRINT " (0) Hercules card"
1110 PRINT " (1) IBM Color Graphics Adapter"
1120 PRINT
1130 INPUT "Select (0 or 1) = ",A
1140 '
1150 X1=55:Y1=0:X2=700:Y2=300:HBIT=9:VBIT=14
1160 IF A=0 THEN GOTO 1220
1170 X1=48:Y1=0:X2=630:Y2=170:HBIT=8:VBIT= 8
1180 IF A=1 THEN GOTO 1220
1190 PRINT "Incorrect input": GOTO 1130
1200 '
1210 ' Size and dimension of drawing:
1220 ' (X1,Y1) = pixel coords for upper left corner
1230 ' (X2,Y2) = pixel coords for lower right corner
1240 '
1250 'HBIT,VBIT :Hor/Vert pixel dimensions for character
1260 '
1270 'Cursor for printing the imageing parameters:
1280 ROW=2+INT(Y1/VBIT):COL=4+INT(X1/HBIT)
1290 '
1300 'Cursor for printing SIN(X)
1310 ROW1=1+INT(Y1/VBIT):COL1=-5+INT(X1/HBIT)
1320 '
1330 '**** Input data ****
1340 '
1350 CLS
1360 INPUT "Set range for abscissa (Angstrom) = ",AP
1370 INPUT "Accelerating voltage (kV) = ",VOLT:
1380 'L=wavelength
1390 PRL=1000*VOLT*(1! + VOLT*1.E-3!*.9788!)
1400 L=12.26!/SQR(PRL)
1410 INPUT "Spherical aberration constant Cs (mm) = ",CS
1420 INPUT "Chromatical aberration constant Cc (mm) = ",CC
1430 '
1440 INPUT "Convergence semiangle (mrad) = ",THTQ
1450 THETA=THTQ*1.E-3!
1460 INPUT "Accelerating voltage fluctuation (dV/V) = ",V
1470 INPUT "Lens current fluctuation (dI/I) = ",I
1480 INPUT "Electron energy spread (dE/E) = ",E
1490 INPUT "Defocus (Angstrom; underfocus negative) = ",DEFOCUS
1500 '
1510 PRINT:PRINT "Electron wavelength = ";L;" ANGSTROM":PRINT
1520 '
1530 PRINT "NOTE: When the plot has been completed,"
1540 PRINT "       press any key to continue": PRINT
1550 PRINT "       *** Press any key to start ***"
```

```
1560 IF INKEY$ ="" THEN GOTO 1560
1570 SCREEN 2:CLS
1580 '
1590 L=12.26!/SQR(PRL): ' L=wavelength
1600 DEL=(CC*10000000!)*SQR(V*V + 4*I*I + E*E)
1610 '
1620 '**** Print imageing parameters ****
1630 '
1640 LOCATE ROW  ,COL:PRINT VOLT;"kV"
1650 LOCATE ROW+1,COL:PRINT "Cs=";CS;"mm"
1660 LOCATE ROW+2,COL:PRINT "Cc=";CC;"mm"
1670 LOCATE ROW+3,COL:PRINT "f = ";DEFOCUS/10;"nm"
1680 LOCATE ROW+4,COL:PRINT "del= ";INT(DEL*.1!+.5!);"nm"
1690 LOCATE ROW+5,COL:PRINT "theta=";THTQ;"mrad"
1700 '
1710 '**** Units on x-axis ****
1720 '
1730 Y0=(Y1+Y2)/2
1740 FOR U=.1! TO (1/(AP)-.05!) STEP .1!
1750 U=INT(10*U+.5!)/10 'to avoid rounding errors
1760 X0=X1 + U*AP*(X2-X1)
1770 I0=2+INT(Y0/VBIT):J0=INT(X0/HBIT):LOCATE I0,J0:PRINT U
1780 LINE(X0,Y0)-(X0,Y0-8)
1790 NEXT U
1800 '
1810 X0=(X1+X2)/2:I0=2+INT(Y2/VBIT):J0=-7+INT(X0/HBIT)
1820 LOCATE I0,J0:PRINT "U (Angstrom -1)"
1830 '
1840 '**** Units on y-axis ****
1850 '
1860 LOCATE ROW1,COL1:PRINT "SIN(X)"
1870 '
1880 J0=-4+INT(X1/HBIT)
1890 '
1900 FOR N=1 TO 7:Y0=Y1+N*(Y2-Y1)/8:LINE (X1,Y0)-(X1+8,Y0):NEXT N
1910 '
1920 Y0=Y1+1*(Y2-Y1)/4:I0=1+INT(Y0/VBIT)
1930 LOCATE I0,J0:PRINT " 0.5"
1940 Y0=Y1+2*(Y2-Y1)/4:I0=1+INT(Y0/VBIT)
1950 LOCATE I0,J0:PRINT " 0.0"
1960 Y0=Y1+3*(Y2-Y1)/4:I0=1+INT(Y0/VBIT)
1970 LOCATE I0,J0:PRINT "-0.5"
1980 '
1990 '**** Draw border of drawing ****
2000 '
2010 LINE (X1,Y1)-(X2,Y1):LINE -(X2,Y2)
2020 LINE -(X1,Y2):LINE -(X1,Y1)
2030 LINE (X2,(Y2+Y1)/2)-(X1,(Y2+Y1)/2):'Home to origon
2040 '
2050 '****** Constants ******
2060 '
2070 PI=3.14159265#
2080 CONA=PI*DEFOCUS*L
2090 CONB=PI*(CS*10000000!)*(L^3)
2100 '
2110 PREL=PI*L*DEL
```

```
2120 CONC=PREL*PREL/2!
2130 '
2140 LN2=LOG(2)
2150 U0=THETA/(L*SQR(LN2))
2160 '
2170 '********* Main loop *********
2180 '
2190 FOR X=X1 TO X2
2200 U=(X-X1)/((X2-X1)*AP):' U(1/D) corr to point X
2210 U2=U*U:U3=U2*U:U4=U3*U
2220 A1=SIN(CONA*U2 + CONB*U4/2):'* Defocus
2230 A2=EXP(-CONC*U4):'           * Chromatic abberation
2240 PW=U0*CONA*U + U0*CONB*U3:'  * Beam divergence
2250 A3=EXP(-PW*PW)
2260 '
2270 YQ=A2*A3: 'Damping term
2280 IF (YQ<1.E-3!) THEN GOTO 2340: 'If zero, skip rest of plot
2290 YQ=A1*YQ
2300 Y=(Y1+Y2)/2 - (Y2-Y1)*YQ/2
2310 LINE -(X,Y)
2320 NEXT X
2330 '
2340 '******* End MAIN loop *******
2350 '
2360 BEEP
2370 IF INKEY$ <>"" THEN GOTO 2370: 'remove previous key(s)
2380 IF INKEY$ ="" THEN GOTO 2380:  'wait for key press
2390 SCREEN 0:CLS
2400 INPUT "Another defocus value (Y/N)";A$
2410 IF A$="Y" OR A$="y" OR A$="" THEN GOTO 1490
2420 INPUT "Another plot for this instrument (Y/N)";A$
2430 IF A$="Y" OR A$="y" OR A$="" THEN GOTO 1440
2440 INPUT "Run program once more (Y/N)";A$
2450 IF A$="N" OR A$="n" THEN SYSTEM
2460 GOTO 1300
```

APPENDIX 2. Examples of output from program CTF.BAS Ver 1.1


SIN(X)
200 kV
Cs= 2.3 mm
Cc= 2.2 mm
f = -60 nm
del= 5 nm
theta= .2 mrad
0.5
0.0
-0.5
.1
.2
.3
.4
.5
.6
.7
U (Angstrom -1)


SIN(X)
200 kV
Cs= 2.3 mm
Cc= 2.2 mm
f = -120 nm
del= 5 nm
theta= .2 mrad
0.5
0.0
-0.5
.1
.2
.3
.4
.5
.6
.7
U (Angstrom -1)

FYSISK INSTITUTTS 8 FORSKNINGS-GRUPPER	DEPARTMENT OF PHYSICS RESEARCH SECTIONS
Allmennfysikk og didaktikk	General Physics
Biofysikk	Biophysics
Elektronikk	Electronics
Elementærpartikkel-fysikk	Experimental Elementary Particle physics
Faste stoffers fysikk	Condensed Matter physics
Kjernefysikk	Nuclear physics
Plasma-, molekylar- og kosmisk fysikk	Plasma-, Molecular and Cosmic physics
Teoretisk fysikk	Theoretical physics

ISBN